F07BHF (SGBRFS/DGBRFS) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07BHF (SGBRFS/DGBRFS) returns error bounds for the solution of a real band system of linear equations with multiple right-hand sides, AX = B or $A^T X = B$. It improves the solution by iterative refinement, in order to reduce the backward error as much as possible.

2 Specification

SUBROUTINE	F07BHF(TRANS, N, KL, KU, NRHS, AB, LDAB, AFB, LDAFB,
1	IPIV, B, LDB, X, LDX, FERR, BERR, WORK, IWORK,
2	INFO)
ENTRY	sgbrfs(TRANS, N, KL, KU, NRHS, AB, LDAB, AFB, LDAFB,
1	IPIV, B, LDB, X, LDX, FERR, BERR, WORK, IWORK,
2	INFO)
INTEGER	N, KL, KU, NRHS, LDAB, LDAFB, IPIV(*), LDB, LDX,
1	IWORK(*), INFO
real	AB(LDAB,*), AFB(LDAFB,*), B(LDB,*), X(LDX,*),
1	<pre>FERR(*), BERR(*), WORK(*)</pre>
CHARACTER*1	TRANS

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine returns the backward errors and estimated bounds on the forward errors for the solution of a real band system of linear equations with multiple right-hand sides AX = B or $A^T X = B$. The routine handles each right-hand side vector (stored as a column of the matrix B) independently, so we describe the function of the routine in terms of a single right-hand side b and solution x.

Given a computed solution x, the routine computes the *component-wise backward error* β . This is the size of the smallest relative perturbation in each element of A and b such that x is the exact solution of a perturbed system

 $(A + \delta A)x = b + \delta b$

 $|\delta a_{ij}| \leq \beta |a_{ij}| \text{ and } |\delta b_i| \leq \beta |b_i|.$

Then the routine estimates a bound for the *component-wise forward error* in the computed solution, defined by:

 $\max_{i} |x_i - \hat{x}_i| / \max_{i} |x_i|$

where \hat{x} is the true solution.

For details of the method, see the the Chapter Introduction.

4 References

 Golub G H and van Loan C F (1996) Matrix Computations Johns Hopkins University Press (3rd Edition), Baltimore

5 Parameters

1: TRANS — CHARACTER*1

On entry: indicates the form of the linear equations for which X is the computed solution as follows:

Input

if TRANS = 'N', then the linear equations are of the form AX = B. if TRANS = 'T' or 'C', then the linear equations are of the form $A^T X = B$.

Constraint: TRANS = 'N', 'T' or 'C'.

	Constraint: $TRANS = 'N'$, 'T' or 'C'.	
2:	N — INTEGER	Input
	On entry: n , the order of the matrix A .	
	Constraint: $N \ge 0$.	
3:	KL — INTEGER	Input
	On entry: k_l , the number of sub-diagonals within the band of A.	1
	Constraint: $KL \ge 0$.	
4:	KU — INTEGER	Input
4.	On entry: k_u , the number of super-diagonals within the band of A.	mput
	Constraint: $KU \ge 0$.	
		-
5:	NRHS — INTEGER	Input
	On entry: r , the number of right-hand sides.	
	Constraint: NRHS ≥ 0 .	
6:	AB(LDAB,*) - real array	Input
	Note: the second dimension of the array AB must be at least $\max(1,N)$.	
	On entry: the n by n original band matrix A as supplied to F07BDF (SGBTRF/DGBTRF stored in rows 1 to $(k_l + k_u + 1)$ of the array rather than in rows $(k_l + 1)$ to $(2k_l + k_u + 1)$.), but
7:	LDAB - INTEGER	Input
	On entry: the first dimension of the array AB as declared in the (sub)program from which F0 (SGBRFS/DGBRFS) is called.	7BHF
	Constraint: $LDAB \ge KL + KU + 1.$	
8:	AFB(LDAFB,*) - real array	Input
	Note: the second dimension of the array AFB must be at least $\max(1,N)$.	
	On entry: the LU factorization of A, as returned by F07BDF (SGBTRF/DGBTRF).	
9:	LDAFB — INTEGER	Input
	On entry: the first dimension of the array AFB as declared in the (sub)program from which F0	7BHF
	(SGBRFS/DGBRFS) is called.	
	Constraint: $LDAFB \ge 2 \times KL + KU + 1.$	
10:	IPIV(*) - INTEGER array	Input
	Note: the dimension of the array IPIV must be at least $\max(1,N)$.	
	On entry: the pivot indices, as returned by F07BDF (SGBTRF/DGBTRF).	
11:	B(LDB,*) - real array	Input
	Note: the second dimension of the array B must be at least $\max(1, \text{NRHS})$.	
	On entry: the n by r right-hand side matrix B .	
12:	LDB — INTEGER	Input
	On entry: the first dimension of the array B as declared in the (sub)program from which F0 (SGBRFS/DGBRFS) is called.	7BHF
	Constraint: LDB $\geq \max(1,N)$.	

13:	X(LDX,*) - real array	Input/Output
	Note: the second dimension of the array X must be at least $\max(1, \text{NRHS})$.	
	On entry: the n by r solution matrix X, as returned by F07BEF (SGBTRS/DGBT	RS).
	On exit: the improved solution matrix X .	
14:	LDX — INTEGER	Input
	On entry: the first dimension of the array X as declared in the (sub)program from v (SGBRFS/DGBRFS) is called.	which F07BHF
	Constraint: LDX $\geq \max(1,N)$.	
15:	$\mathrm{FERR}(*) - \boldsymbol{real} \; \mathrm{array}$	Output
	Note: the dimension of the array FERR must be at least $\max(1, \text{NRHS})$.	
	On exit: $\text{FERR}(j)$ contains an estimated error bound for the <i>j</i> th solution vector, t column of X, for $j = 1, 2,, r$.	hat is, the j th
16:	$\operatorname{BERR}(*)$ — <i>real</i> array	Output
	Note: the dimension of the array BERR must be at least $\max(1, \text{NRHS})$.	
	On exit: BERR(j) contains the component-wise backward error bound β for the jth s that is, the jth column of X, for $j = 1, 2,, r$.	olution vector,
17:	WORK(*) - real array	Work space
	Note: the dimension of the array WORK must be at least $\max(1,3*N)$.	
18:	IWORK(*) - INTEGER array	Work space
	Note: the dimension of the array IWORK must be at least $\max(1,N)$.	
19:	INFO — INTEGER	Output
	On exit: INFO = 0 unless the routine detects an error (see Section 6).	

6 Error Indicators and Warnings

INFO < 0

If INFO = -i, the *i*th parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The bounds returned in FERR are not rigorous, because they are estimated, not computed exactly; but in practice they almost always overestimate the actual error.

8 Further Comments

For each right-hand side, computation of the backward error involves a minimum of $4n(k_l + k_u)$ floatingpoint operations. Each step of iterative refinement involves an additional $2n(4k_l + 3k_u)$ operations. This assumes $n \gg k_l$ and $n \gg k_u$. At most 5 steps of iterative refinement are performed, but usually only 1 or 2 steps are required.

Estimating the forward error involves solving a number of systems of linear equations of the form Ax = b or $A^Tx = b$; the number is usually 4 or 5 and never more than 11. Each solution involves approximately $2n(2k_l + k_u)$ operations.

The complex analogue of this routine is F07BVF (CGBRFS/ZGBRFS).

9 Example

To solve the system of equations AX = B using iterative refinement and to compute the forward and backward error bounds, where

$$A = \begin{pmatrix} -0.23 & 2.54 & -3.66 & 0.00 \\ -6.98 & 2.46 & -2.73 & -2.13 \\ 0.00 & 2.56 & 2.46 & 4.07 \\ 0.00 & 0.00 & -4.78 & -3.82 \end{pmatrix} \text{ and } B = \begin{pmatrix} 4.42 & -36.01 \\ 27.13 & -31.67 \\ -6.14 & -1.16 \\ 10.50 & -25.82 \end{pmatrix}.$$

Here A is nonsymmetric and is treated as a band matrix, which must first be factorized by F07BDF (SGBTRF/DGBTRF).

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
FO7BHF Example Program Text
*
*
     Mark 15 Release. NAG Copyright 1991.
      .. Parameters ..
*
     INTEGER
                      NIN, NOUT
     PARAMETER
                      (NIN=5,NOUT=6)
     real
                       ZERO
     PARAMETER
                       (ZERO=0.0e0)
                       NMAX, NRHMAX, KLMAX, KUMAX, LDAB, LDAFB, LDB, LDX
     INTEGER
     PARAMETER
                       (NMAX=8, NRHMAX=NMAX, KLMAX=8, KUMAX=8,
     +
                       LDAB=KLMAX+KUMAX+1,LDAFB=2*KLMAX+KUMAX+1,
                       LDB=NMAX,LDX=NMAX)
     +
     CHARACTER
                       TRANS
                       (TRANS='N')
     PARAMETER
      .. Local Scalars ..
      INTEGER
                       I, IFAIL, INFO, J, K, KL, KU, N, NRHS
      .. Local Arrays ..
                       AB(LDAB,NMAX), AFB(LDAFB,NMAX), B(LDB,NRHMAX),
     real
                       BERR(NRHMAX), FERR(NRHMAX), WORK(3*NMAX),
     +
                       X(LDX,NMAX)
     +
     INTEGER
                       IPIV(NMAX), IWORK(NMAX)
      .. External Subroutines ..
                      F06QFF, F06QHF, sgbrfs, sgbtrf, sgbtrs, X04CAF
     EXTERNAL
      .. Intrinsic Functions ...
     INTRINSIC
                     MAX, MIN
      .. Executable Statements ..
     WRITE (NOUT,*) 'F07BHF Example Program Results'
     Skip heading in data file
*
     READ (NIN,*)
     READ (NIN,*) N, NRHS, KL, KU
     IF (N.LE.NMAX .AND. NRHS.LE.NRHMAX .AND. KL.LE.KLMAX .AND. KU.LE.
         KUMAX) THEN
        Set A to zero to avoid referencing uninitialized elements
*
*
        CALL F06QHF('General',KL+KU+1,N,ZERO,ZERO,AB,LDAB)
        Read A and B from data file, and copy A to AFB and B to X
*
        K = KU + 1
        READ (NIN,*) ((AB(K+I-J,J),J=MAX(I-KL,1),MIN(I+KU,N)),I=1,N)
        READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
```

*

F07BHF (SGBRFS/DGBRFS)

```
CALL F06QFF('General',KL+KU+1,N,AB,LDAB,AFB(KL+1,1),LDAFB)
*
         CALL F06QFF('General', N, NRHS, B, LDB, X, LDX)
*
         Factorize A in the array AFB
*
         CALL sgbtrf(N,N,KL,KU,AFB,LDAFB,IPIV,INFO)
*
         WRITE (NOUT, *)
         IF (INFO.EQ.O) THEN
*
            Compute solution in the array X
*
            CALL sgbtrs(TRANS,N,KL,KU,NRHS,AFB,LDAFB,IPIV,X,LDX,INFO)
*
*
            Improve solution, and compute backward errors and
            estimated bounds on the forward errors
*
*
            CALL sgbrfs(TRANS,N,KL,KU,NRHS,AB,LDAB,AFB,LDAFB,IPIV,B,LDB,
     +
                         X,LDX,FERR,BERR,WORK,IWORK,INFO)
*
            Print solution
*
*
            IFAIL = 0
*
            CALL X04CAF('General',' ',N,NRHS,X,LDX,'Solution(s)',IFAIL)
*
            WRITE (NOUT,*)
            WRITE (NOUT, *) 'Backward errors (machine-dependent)'
            WRITE (NOUT, 99999) (BERR(J), J=1, NRHS)
            WRITE (NOUT,*)
              'Estimated forward error bounds (machine-dependent)'
     +
            WRITE (NOUT, 99999) (FERR(J), J=1, NRHS)
         ELSE
            WRITE (NOUT,*) 'The factor U is singular'
         END IF
      END IF
      STOP
99999 FORMAT ((3X,1P,7e11.1))
      END
```

9.2 Program Data

F07BHF Example Program Data 4 2 1 2 :Values of N, NRHS, KL and KU -0.23 2.54 -3.66 -6.98 2.46 -2.73 -2.13 2.56 2.46 4.07 -4.78 -3.82 :End of matrix A 4.42 -36.01 27.13 -31.67 -6.14 -1.16 10.50 -25.82 :End of matrix B

9.3 Program Results

F07BHF Example Program Results

Solution(s) 1 2 1 -2.0000 1.0000 2 3.0000 -4.0000 3 1.0000 7.0000 4 -4.0000 -2.0000 Backward errors (machine-dependent) 1.1E-16 9.9E-17 Estimated forward error bounds (machine-dependent) 1.6E-14 1.9E-14